contract MBTPool is Ownable, ReentrancyGuard {
    using SafeMath for uint256;
    using SafeERC20 for IERC20;

    // The address of the smart chef factory
    address public SMART_CHEF_FACTORY;
    
    // Platform fee wallet
    address public feeAddress;
    
    // Whether a limit is set for users
    bool public hasUserLimit;

    // Whether it is initialized
    bool public isInitialized;

    // Accrued token per share
    uint256 public accTokenPerShare;

    // The block number when BNNF mining ends.
    uint256 public bonusEndBlock;

    // The block number when BNNF mining starts.
    uint256 public startBlock;

    // The block number of the last pool update
    uint256 public lastRewardBlock;

    // The pool limit (0 if none)
    uint256 public poolLimitPerUser;

    // BNNF tokens created per block.
    uint256 public rewardPerBlock;

    // The precision factor
    uint256 public PRECISION_FACTOR;
    
    uint256 public lockStakeDate;
    uint256 public lockWithdrawDate;
    
    // The reward array
    mapping(uint256 => uint256) public rewardPerMonth;
    uint256 constant blocksPerMonth = 199380;

    // The reward token
    IERC20 public rewardToken;

    // The staked token
    IERC20 public stakedToken;
    

    // Info of each user that stakes tokens (stakedToken)
    mapping(address => UserInfo) public userInfo;

    struct UserInfo {
        uint256 amount; // How many staked tokens the user has provided
        uint256 rewardDebt; // Reward debt
        uint256 withdrawDate; // withdrawDate
    }

    event AdminTokenRecovery(address tokenRecovered, uint256 amount);
    event Deposit(address indexed user, uint256 amount);
    event EmergencyWithdraw(address indexed user, uint256 amount);
    event NewStartAndEndBlocks(uint256 startBlock, uint256 endBlock);
    event NewRewardPerBlock(uint256 rewardPerBlock);
    event NewPoolLimit(uint256 poolLimitPerUser);
    event RewardsStop(uint256 blockNumber);
    event Withdraw(address indexed user, uint256 amount);
    event Initialize(IERC20 _stakedToken, IERC20 _rewardToken, uint256 _rewardPerBlock, 
        uint256 _startBlock, uint256 _bonusEndBlock, uint256 _poolLimitPerUser, address _admin, uint256 _lockStakeDate,
        uint256 _lockWithdrawDate, address _feeAddress);
    event StopReward();

    constructor() {
        SMART_CHEF_FACTORY = msg.sender;
    }

    /*
     * @notice Initialize the contract
     * @param _stakedToken: staked token address
     * @param _rewardToken: reward token address
     * @param _rewardPerBlock: reward per block (in rewardToken)
     * @param _startBlock: start block
     * @param _bonusEndBlock: end block
     * @param _poolLimitPerUser: pool limit per user in stakedToken (if any, else 0)
     * @param _admin: admin address with ownership
     */
    function initialize(
        IERC20 _stakedToken,
        IERC20 _rewardToken,
        uint256 _rewardPerBlock,
        uint256 _startBlock,
        uint256 _bonusEndBlock,
        uint256 _poolLimitPerUser,
        address _admin,
        uint256 _lockStakeDate,
        uint256 _lockWithdrawDate,
        address _feeAddress
    ) external {
        require(!isInitialized, 'Already initialized');
        require(msg.sender == SMART_CHEF_FACTORY, 'Not factory');
        require(_feeAddress != address(0), "_feeAddress: fee address is zero address");
        require(_admin != address(0), "_admin: admin address is zero address");

        // Make this contract initialized
        isInitialized = true;
        
        feeAddress = _feeAddress;

        lockStakeDate = _lockStakeDate;
        lockWithdrawDate = _lockWithdrawDate;
        stakedToken = _stakedToken;
        rewardToken = _rewardToken;
        rewardPerBlock = _rewardPerBlock;
        startBlock = _startBlock;
        bonusEndBlock = _bonusEndBlock;
        
        // reward per block monthly
        rewardPerMonth[0] = 945 * 10 ** 16;
        rewardPerMonth[1] = 84 * 10 ** 17;
        rewardPerMonth[2] = 69 * 10 ** 17;
        rewardPerMonth[3] = 563 * 10 ** 16;
        rewardPerMonth[4] = 36 * 10 ** 17;
        rewardPerMonth[5] = 278 * 10 ** 16;
        rewardPerMonth[6] = 18 * 10 ** 17;
        rewardPerMonth[7] = 98 * 10 ** 16;
        rewardPerMonth[8] = 98 * 10 ** 16;
        rewardPerMonth[9] = 83 * 10 ** 16;
        rewardPerMonth[10] = 83 * 10 ** 16;
        rewardPerMonth[11] = 83 * 10 ** 16;

        if (_poolLimitPerUser > 0) {
            hasUserLimit = true;
            poolLimitPerUser = _poolLimitPerUser;
        }

        uint256 decimalsRewardToken = uint256(rewardToken.decimals());
        require(decimalsRewardToken < 30, 'Must be inferior to 30');

        PRECISION_FACTOR = uint256(10**(uint256(30).sub(decimalsRewardToken)));

        // Set the lastRewardBlock as the startBlock
        lastRewardBlock = startBlock;

        // Transfer ownership to the admin address who becomes owner of the contract
        transferOwnership(_admin);

        emit Initialize(_stakedToken, _rewardToken, _rewardPerBlock, _startBlock, _bonusEndBlock, _poolLimitPerUser,
            _admin, _lockStakeDate, _lockWithdrawDate, _feeAddress);
    }

    /*
     * @notice Deposit staked tokens and collect reward tokens (if any)
     * @param _amount: amount to withdraw (in rewardToken)
     */
    function deposit(uint256 _amount) external nonReentrant {
        UserInfo storage user = userInfo[msg.sender];

        if (hasUserLimit) {
            require(_amount.add(user.amount) <= poolLimitPerUser, 'User amount above limit');
        }
        
        if (lockStakeDate != 0 && _amount > 0) {
            require(block.timestamp < lockStakeDate, 'BANANA: stake locked');
        }

        _updatePool();

        if (user.amount > 0) {
            uint256 pending = user.amount.mul(accTokenPerShare).div(PRECISION_FACTOR).sub(user.rewardDebt);
            if (pending > 0) {
                uint256 fee = pending * 3 / 1000;
                rewardToken.safeTransfer(address(msg.sender), pending - fee);
                rewardToken.safeTransfer(address(feeAddress), fee);
            }
        }

        if (_amount > 0) {
            user.amount = user.amount.add(_amount);
            stakedToken.safeTransferFrom(address(msg.sender), address(this), _amount);
            
            user.withdrawDate = block.timestamp;
        }

        user.rewardDebt = user.amount.mul(accTokenPerShare).div(PRECISION_FACTOR);

        emit Deposit(msg.sender, _amount);
    }

    /*
     * @notice Withdraw staked tokens and collect reward tokens
     * @param _amount: amount to withdraw (in rewardToken)
     */
    function withdraw(uint256 _amount) external nonReentrant {
        UserInfo storage user = userInfo[msg.sender];
        require(user.amount >= _amount, 'Amount to withdraw too high');
        require(user.withdrawDate + lockWithdrawDate >= block.timestamp, 'BANANA: Withdraw was locked');

        _updatePool();

        uint256 pending = user.amount.mul(accTokenPerShare).div(PRECISION_FACTOR).sub(user.rewardDebt);

        if (_amount > 0) {
            user.amount = user.amount.sub(_amount);
            stakedToken.safeTransfer(address(msg.sender), _amount);
        }

        if (pending > 0) {
            uint256 fee = pending * 3 / 1000;
            rewardToken.safeTransfer(address(msg.sender), pending - fee);
            rewardToken.safeTransfer(address(feeAddress), fee);
        }

        user.rewardDebt = user.amount.mul(accTokenPerShare).div(PRECISION_FACTOR);
        
        emit Withdraw(msg.sender, _amount);
    }

    /*
     * @notice Withdraw staked tokens without caring about rewards rewards
     * @dev Needs to be for emergency.
     */
    function emergencyWithdraw() external nonReentrant {
        UserInfo storage user = userInfo[msg.sender];
        uint256 amountToTransfer = user.amount;
        user.amount = 0;
        user.rewardDebt = 0;

        if (amountToTransfer > 0) {
            stakedToken.safeTransfer(address(msg.sender), amountToTransfer);
        }

        emit EmergencyWithdraw(msg.sender, user.amount);
    }

    /*
     * @notice Stop rewards
     * @dev Only callable by owner. Needs to be for emergency.
     */
    function emergencyRewardWithdraw(uint256 _amount) external onlyOwner {
        rewardToken.safeTransfer(address(msg.sender), _amount);
        emit EmergencyWithdraw(msg.sender, _amount);
    }

    /**
     * @notice It allows the admin to recover wrong tokens sent to the contract
     * @param _tokenAddress: the address of the token to withdraw
     * @param _tokenAmount: the number of tokens to withdraw
     * @dev This function is only callable by admin.
     */
    function recoverWrongTokens(address _tokenAddress, uint256 _tokenAmount) external onlyOwner {
        require(_tokenAddress != address(stakedToken), 'Cannot be staked token');
        require(_tokenAddress != address(rewardToken), 'Cannot be reward token');

        IERC20(_tokenAddress).safeTransfer(address(msg.sender), _tokenAmount);

        emit AdminTokenRecovery(_tokenAddress, _tokenAmount);
    }

    /*
     * @notice Stop rewards
     * @dev Only callable by owner
     */
    function stopReward() external onlyOwner {
        bonusEndBlock = block.number;
        emit StopReward();
    }

    /*
     * @notice Update pool limit per user
     * @dev Only callable by owner.
     * @param _hasUserLimit: whether the limit remains forced
     * @param _poolLimitPerUser: new pool limit per user
     */
    function updatePoolLimitPerUser(bool _hasUserLimit, uint256 _poolLimitPerUser) external onlyOwner {
        if (_hasUserLimit) {
            require(_poolLimitPerUser > poolLimitPerUser, 'New limit must be higher');
            poolLimitPerUser = _poolLimitPerUser;
        } else {
            hasUserLimit = _hasUserLimit;
            poolLimitPerUser = 0;
        }
        emit NewPoolLimit(poolLimitPerUser);
    }

    /*
     * @notice Update reward per block
     * @dev Only callable by owner.
     * @param _rewardPerBlock: the reward per block
     */
    function updateRewardPerBlock(uint256 _rewardPerBlock) external onlyOwner {
        require(block.number < startBlock, 'Pool has started');
        rewardPerBlock = _rewardPerBlock;
        emit NewRewardPerBlock(_rewardPerBlock);
    }

    /**
     * @notice It allows the admin to update start and end blocks
     * @dev This function is only callable by owner.
     * @param _startBlock: the new start block
     * @param _bonusEndBlock: the new end block
     */
    function updateStartAndEndBlocks(uint256 _startBlock, uint256 _bonusEndBlock) external onlyOwner {
        require(block.number < startBlock, 'Pool has started');
        require(_startBlock < _bonusEndBlock, 'New startBlock must be lower than new endBlock');
        require(block.number < _startBlock, 'New startBlock must be higher than current block');

        startBlock = _startBlock;
        bonusEndBlock = _bonusEndBlock;

        // Set the lastRewardBlock as the startBlock
        lastRewardBlock = startBlock;

        emit NewStartAndEndBlocks(_startBlock, _bonusEndBlock);
    }

    /*
     * @notice View function to see pending reward on frontend.
     * @param _user: user address
     * @return Pending reward for a given user
     */
    function pendingReward(address _user) external view returns (uint256) {
        UserInfo storage user = userInfo[_user];
        uint256 stakedTokenSupply = stakedToken.balanceOf(address(this));
        if (block.number > lastRewardBlock && stakedTokenSupply != 0) {
            uint256 bnnfReward = _getRewardBNN(lastRewardBlock, block.number);
            uint256 adjustedTokenPerShare = accTokenPerShare.add(
                bnnfReward.mul(PRECISION_FACTOR).div(stakedTokenSupply)
            );
            return user.amount.mul(adjustedTokenPerShare).div(PRECISION_FACTOR).sub(user.rewardDebt);
        } else {
            return user.amount.mul(accTokenPerShare).div(PRECISION_FACTOR).sub(user.rewardDebt);
        }
    }

    /*
     * @notice Update reward variables of the given pool to be up-to-date.
     */
    function _updatePool() internal {
        if (block.number <= lastRewardBlock) {
            return;
        }

        uint256 stakedTokenSupply = stakedToken.balanceOf(address(this));

        if (stakedTokenSupply > 0) {
            uint256 bnnfReward = _getRewardBNN(lastRewardBlock, block.number);
            accTokenPerShare = accTokenPerShare.add(bnnfReward.mul(PRECISION_FACTOR).div(stakedTokenSupply));
        }

        lastRewardBlock = block.number;
    }
    
    function getLockStakeDate() external view returns (uint256) {
        return lockStakeDate; 
    }
    
    function getLockWithdrawDate() external view returns (uint256) {
        UserInfo storage user = userInfo[msg.sender];
        return user.withdrawDate + lockWithdrawDate;
    }
    
    /*
     * @notice View function to get bnn reward on frontend.
     * @param _from: from address
     * @param _to: to address
     * @return bnn reward
     */
    function _getRewardBNN(uint256 _from, uint256 _to) internal view returns (uint256) {
        if (_to <= bonusEndBlock) {
            uint256 fromMonth = (_from - startBlock) / blocksPerMonth;
            uint256 toMonth = (_to - startBlock) / blocksPerMonth;
            
            if (fromMonth == toMonth) {
                return rewardPerMonth[fromMonth] * _to.sub(_from);
            }
            
            uint256 reward = 0;
            for(uint256 i = fromMonth; i <= toMonth; i++) {
                if (i == fromMonth) {
                    reward += rewardPerMonth[i % 12] *(blocksPerMonth - _from);
                } else if (i != toMonth) {
                    reward += rewardPerMonth[i % 12] * blocksPerMonth;
                } else {
                    reward += rewardPerMonth[i % 12] * (_to - startBlock - blocksPerMonth * (i - 1));
                }
            }
            
            return reward;
        } else if (_from >= bonusEndBlock) {
            return 0;
        } else {
            _to = bonusEndBlock;
            uint256 fromMonth = (_from - startBlock) / blocksPerMonth;
            uint256 toMonth = (_to - startBlock) / blocksPerMonth;
            
            if (fromMonth == toMonth) {
                return rewardPerMonth[fromMonth] * _to.sub(_from);
            }
            
            uint256 reward = 0;
            for(uint256 i = fromMonth; i <= toMonth; i++) {
                if (i == fromMonth) {
                    reward += rewardPerMonth[i % 12] *(blocksPerMonth - _from);
                } else if (i != toMonth) {
                    reward += rewardPerMonth[i % 12] * blocksPerMonth;
                } else {
                    reward += rewardPerMonth[i % 12] * (_to - startBlock - blocksPerMonth * (i - 1));
                }
            }
            return reward;
        }
    }
}
function updatePoolLimitPerUser(bool _hasUserLimit, uint256 _poolLimitPerUser) external onlyOwner {
        if (_hasUserLimit) {
            require(_poolLimitPerUser > poolLimitPerUser, 'New limit must be higher');
            poolLimitPerUser = _poolLimitPerUser;
        } else {
            hasUserLimit = _hasUserLimit;
            poolLimitPerUser = 0;
        }
        emit NewPoolLimit(poolLimitPerUser);
    }
/*
 * @notice Update pool limit per user
 * @dev Only callable by owner.
 * @param _hasUserLimit: whether the limit remains forced
 * @param _poolLimitPerUser: new pool limit per user
 */